
PROC ANOVA and PROC GLM 
 
 
 
OPTIONS NOCENTER NODATE NONUMBER LS=80 PS=40 MISSING='.'; 
 
******************************************************************* 
*PROGRAM TO ILLUSTRATE USE OF PROC ANOVA, PROC GLM – ANALYSIS OF  
*VARIANCE IS USED TO TEST THE DIFFERENCE BETWEEN MEAN TEST GRADES  
*OF STUDENTS ON A MATH EXAM TAUGHT BY ONE OF 3 DIFFERENT PROFESSORS 
* AND HAVING EITHER 1 OR 2 YEARS OF CALCULUS UNDER THEIR BELTS. 
*LJ GORMEZANO --CREATED 3/6/05 
*******************************************************************; 
 
DATA MATH; 

INPUT CALC 1 PROF $3 GRADE 5-7; 
CARDS; 

1 A 65 
1 B 70 
1 C 90 
1 B 85 
1 B 95 
1 C 100 
1 A 75 
1 C 89 
2 B 87 
2 A 73 
2 C 75 
2 A 55 
2 B 79 
2 C 98 
2 A 82 
; 
 
PROC SORT DATA=MATH; 
 BY CALC PROF; 
RUN; 
 
*ONE-WAY ANOVA TO TEST THE EFFECT OF HAVING A DIFFERENT PROFESSOR ON MEAN 
TEST GRADES; 
PROC ANOVA DATA=MATH; 
 CLASS PROF; 
 MODEL GRADE=PROF; 
RUN; 

*ONE-WAY ANOVA TO TEST THE EFFECT OF PROFESSOR ON MEAN TEST GRADES; 
PROC ANOVA DATA=MATH; 
 CLASS PROF; 
 MODEL GRADE=PROF; 
 MEANS PROF/LSD BON; 
RUN; 

*TWO-WAY ANOVA USING PROC GLM TO TEST THE EFFECT OF PROFESSOR ON 
MEAN TEST GRADES; 
PROC GLM DATA=MATH; 
 CLASS CALC PROF; 
 MODEL GRADE=CALC PROF CALC*PROF; 
 LSMEANS CALC PROF CALC*PROF/ADJUST=BON TDIFF PDIFF; 
 CONTRAST 'A VS. C' PROF 1 0 -1; 
 CONTRAST 'A VS. B' PROF 1 -1 0; 
 CONTRAST 'B VS. C' PROF 0 1 -1; 
RUN; 

